



A surface of degree 24 with 1440 singularities of type D_4

Cédric Bonnafé

► To cite this version:

| Cédric Bonnafé. A surface of degree 24 with 1440 singularities of type D_4 . 2018. hal-01773229v2

HAL Id: hal-01773229

<https://hal.science/hal-01773229v2>

Preprint submitted on 23 Apr 2018 (v2), last revised 3 Jul 2018 (v4)

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

A SURFACE OF DEGREE 24 WITH 1440 SINGULARITIES OF TYPE D_4

by

CÉDRIC BONNAFÉ

Abstract. — Using the invariant algebra of the reflection group denoted by G_{32} in Shephard-Todd classification, we construct three irreducible surfaces in $\mathbf{P}^3(\mathbb{C})$ with many singularities: one of them has degree 24 and contains 1440 quotient singularities of type D_4 .

Let $\mu_{D_4}(d)$ denote the maximal number of quotient singularities of type D_4 that an irreducible projective surface of degree d in $\mathbf{P}^3(\mathbb{C})$ might have. Miyaoka [Miy] proved that

$$\mu_{D_4}(d) \leq \frac{16}{117}d(d-1)^2$$

For $d = 8, 16$ or 24 , this reads

$$\mu_{D_4}(8) \leq 53, \quad \mu_{D_4}(16) \leq 492 \quad \text{and} \quad \mu_{D_4}(24) \leq 1736.$$

The main results of this note are that

$$(1) \quad \mu_{D_4}(8) \geq 44, \quad \mu_{D_4}(16) \geq 352 \quad \text{and} \quad \mu_{D_4}(24) \geq 1440$$

and that

$$(2) \quad \mu_{D_4}(8k) \geq 44k^3$$

for all $k \geq 1$. So, let $\mathbb{C}[x_1, x_2, x_3, x_4]$ be the polynomial ring over $\mathbb{C}$ in 4 indeterminates with its usual grading and let $\mathbf{P}^3(\mathbb{C}) = \text{Proj}(\mathbb{C}[x_1, x_2, x_3, x_4])$ be the associated projective space of dimension 3. If $f \in \mathbb{C}[x_1, x_2, x_3, x_4]$ is homogeneous, we denote by $\mathcal{Z}(f)$ the projective hypersurface it defines. If k is a natural number, we denote by $f[k]$ the homogeneous polynomial $f(x_1^k, x_2^k, x_3^k, x_4^k)$. We aim to prove the following result:

Theorem. *There exists an homogenous polynomial g of degree 8 such that:*

- (a) $\mathcal{Z}(g)$ is an irreducible surface of degree 8 in $\mathbf{P}^3(\mathbb{C})$ with exactly 44 singular points which are all quotient singularities of type D_4 .
- (b) If $k \geq 1$, then $\mathcal{Z}(g[k])$ is an irreducible surface of degree $8k$ with at least $44k^3$ quotient singularities of type D_4 .
- (c) $\mathcal{Z}(g[2])$ is an irreducible surface of degree 16 with exactly 472 singular points: 24 quotient singularities of type A_1 , 96 quotient singularities of type A_2 and 352 quotient singularities of type D_4 .
- (d) $\mathcal{Z}(g[3])$ is an irreducible surface of degree 24 in $\mathbf{P}^3(\mathbb{C})$ with exactly 1440 singular points which are all quotient singularities of type D_4 . The automorphism group of $\mathcal{Z}(g[3])$ contains at least 25920 elements and acts transitively on the 1440 singular points.

The proof relies on extensive MAGMA computations [Magma], and goes as follow. Let

$$\begin{aligned}
g = & x_1^8 - 6x_1^6x_2^2 - 60x_1^6x_2x_3 - 60x_1^6x_2x_4 - 6x_1^6x_3^2 + 60x_1^6x_3x_4 - 6x_1^6x_4^2 \\
& + 2240x_1^5x_2^2x_3 - 2240x_1^5x_2^2x_4 + 2240x_1^5x_2x_3^2 - 2240x_1^5x_2x_4^2 + 2240x_1^5x_3^2x_4 \\
& - 2240x_1^5x_3x_4^2 - 14x_1^4x_2^4 + 10180x_1^4x_2^3x_3 + 10180x_1^4x_2^3x_4 + 40412x_1^4x_2^2x_3^2 \\
& - 23440x_1^4x_2^2x_3x_4 + 40412x_1^4x_2^2x_4^2 + 10180x_1^4x_2x_3^3 + 23440x_1^4x_2x_3^2x_4 \\
& + 23440x_1^4x_2x_3x_4^2 + 10180x_1^4x_2x_4^3 - 14x_1^4x_3^4 - 10180x_1^4x_3^3x_4 + 40412x_1^4x_3^2x_4^2 \\
& - 10180x_1^4x_3x_4^3 - 14x_1^4x_4^4 + 10180x_1^3x_2^4x_3 - 10180x_1^3x_2^4x_4 + 111980x_1^3x_2^3x_3^2 \\
& - 111980x_1^3x_2^3x_4^2 + 111980x_1^3x_2^2x_3^3 - 111980x_1^3x_2^2x_4^3 + 10180x_1^3x_2x_3^4 - 10180x_1^3x_2x_4^4 \\
& + 10180x_1^3x_3^4x_4 - 111980x_1^3x_3^3x_2^2 + 111980x_1^3x_3^2x_4^3 - 10180x_1^3x_3x_4^4 - 6x_1^2x_2^6 \\
& + 2240x_1^2x_2^5x_3 + 2240x_1^2x_2^5x_4 + 40412x_1^2x_2^4x_3^2 + 23440x_1^2x_2^4x_3x_4 + 40412x_1^2x_2^4x_4^2 \\
& + 111980x_1^2x_2^3x_3^3 + 111980x_1^2x_2^3x_4^3 + 40412x_1^2x_2^2x_3^4 + 154704x_1^2x_2^2x_3^2x_4^2 + 40412x_1^2x_2^2x_4^4 \\
& + 2240x_1^2x_2x_3^5 - 23440x_1^2x_2x_3^4x_4 - 23440x_1^2x_2x_3x_4^4 + 2240x_1^2x_2x_4^5 - 6x_1^2x_3^6 \\
& - 2240x_1^2x_3^5x_4 + 40412x_1^2x_3^4x_2^2 - 111980x_1^2x_3^3x_4^3 + 40412x_1^2x_3^2x_4^4 - 2240x_1^2x_3x_4^5 \\
& - 6x_1^2x_4^6 - 60x_1x_2^6x_3 + 60x_1x_2^6x_4 + 2240x_1x_2^5x_3^2 - 2240x_1x_2^5x_4^2 + 10180x_1x_2^4x_3^3 \\
& - 23440x_1x_2^4x_3^2x_4 + 23440x_1x_2^4x_3x_4^2 - 10180x_1x_2^4x_4^3 + 10180x_1x_2^3x_4^4 - 10180x_1x_2^3x_4^4 \\
& + 2240x_1x_2^2x_3^5 + 23440x_1x_2^2x_3^4x_4 - 23440x_1x_2^2x_3x_4^4 - 2240x_1x_2^2x_4^5 - 60x_1x_2x_3^6 \\
& + 23440x_1x_2x_3^4x_4^2 - 23440x_1x_2x_3^2x_4^4 + 60x_1x_2x_4^6 - 60x_1x_3^6x_4 - 2240x_1x_3^5x_4^2 \\
& + 10180x_1x_3^4x_4^3 - 10180x_1x_3^3x_4^4 + 2240x_1x_3^2x_4^5 + 60x_1x_3x_4^6 + x_2^8 - 6x_2^6x_3^2 \\
& - 60x_2^6x_3x_4 - 6x_2^6x_4^2 - 2240x_2^5x_3^2x_4 - 2240x_2^5x_3x_4^2 - 14x_2^4x_3^4 + 10180x_2^4x_3^3x_4 \\
& + 40412x_2^4x_3^2x_4^2 + 10180x_2^4x_3x_4^3 - 14x_2^4x_4^4 - 10180x_2^3x_3^4x_4 - 111980x_2^3x_3^3x_4^2 \\
& - 111980x_2^3x_3^2x_4^3 - 10180x_2^3x_3x_4^4 - 6x_2^2x_3^6 + 2240x_2^2x_3^5x_4 + 40412x_2^2x_3^4x_4^2 \\
& + 111980x_2^2x_3^3x_4^3 + 40412x_2^2x_3^2x_4^4 + 2240x_2^2x_3x_4^5 - 6x_2^2x_4^6 + 60x_2x_3^6x_4 \\
& - 2240x_2x_3^5x_4^2 - 10180x_2x_3^4x_4^3 - 10180x_2x_3^3x_4^4 - 2240x_2x_3^2x_4^5 + 60x_2x_3x_4^6 \\
& + x_3^8 - 6x_3^6x_4^2 - 14x_3^4x_4^4 - 6x_3^2x_4^6 + x_4^8.
\end{aligned}$$

This is the polynomial of the Theorem. It has coefficients in $\mathbb{Q}$ but note for instance that the singular points of $\mathcal{Z}(g)$ and $\mathcal{Z}(g[3])$ have coordinates in the cyclotomic field $K = \mathbb{Q}(z)$, where z is a primitive 12-th root of unity. We denote by $\mathcal{Z}(g)_{\text{sing}}$ the (reduced) singular locus of $\mathcal{Z}(g)$ respectively.

Remark 1. As we will see in the computations below, the polynomial $g[3]$ is invariant under the action of the complex reflection group denoted by G_{32} in Shephard-Todd classification. In fact, we did not find g directly: we found first $g[3]$ by looking at invariants of degree 24 of G_{32} , following ideas of Barth [Bar] (for constructing *Barth sextic* with 65 nodes) and Sarti [Sar1], [Sar2], [Sar3] (for constructing for instance the *Sarti surface* of degree 12 with 600 nodes), who used invariants of Coxeter groups of type H_3 and H_4 . ■

Remark 2. As the expert in computer algebra will notice by reading the MAGMA codes below, the author is not a specialist in programming and asks for some indulgence. ■

So let us start by defining the polynomial g in MAGMA. We will define it over the rational numbers. This will not be important for the computations regarding $\mathcal{Z}(g)$, but this will shorten drastically computations regarding $\mathcal{Z}(g[3])$.

```

> Q:=RationalField();
> P3<x1,x2,x3,x4>:=ProjectiveSpace(Q,3);
>
> g:=x1^8 - 6*x1^6*x2^2 - 60*x1^6*x2*x3
> - 60*x1^6*x2*x4 - 6*x1^6*x3^2 + 60*x1^6*x3*x4
> - 6*x1^6*x4^2 + 2240*x1^5*x2^2*x3
> - 2240*x1^5*x2^2*x4 + 2240*x1^5*x2*x3^2
> - 2240*x1^5*x2*x4^2 + 2240*x1^5*x3^2*x4
> - 2240*x1^5*x3*x4^2 - 14*x1^4*x2^4
> + 10180*x1^4*x2^3*x3 + 10180*x1^4*x2^3*x4
> + 40412*x1^4*x2^2*x3^2 - 23440*x1^4*x2^2*x3*x4
> + 40412*x1^4*x2^2*x4^2 + 10180*x1^4*x2*x3^3
> + 23440*x1^4*x2*x3^2*x4 + 23440*x1^4*x2*x3*x4^2
> + 10180*x1^4*x2*x4^3 - 14*x1^4*x3^4
> - 10180*x1^4*x3^3*x4 + 40412*x1^4*x3^2*x4^2
> - 10180*x1^4*x3*x4^3 - 14*x1^4*x4^4
> + 10180*x1^3*x2^4*x3 - 10180*x1^3*x2^4*x4
> + 111980*x1^3*x2^3*x3^2 - 111980*x1^3*x2^3*x4^2
> + 111980*x1^3*x2^2*x3^3 - 111980*x1^3*x2^2*x4^3
> + 10180*x1^3*x2*x3^4 - 10180*x1^3*x2*x4^4
> + 10180*x1^3*x3^4*x4 - 111980*x1^3*x3^3*x4^2
> + 111980*x1^3*x3^2*x4^3 - 10180*x1^3*x3*x4^4
> - 6*x1^2*x2^6 + 2240*x1^2*x2^5*x3 + 2240*x1^2*x2^5*x4
> + 40412*x1^2*x2^4*x3^2 + 23440*x1^2*x2^4*x3*x4
> + 40412*x1^2*x2^4*x4^2 + 111980*x1^2*x2^3*x3^3
> + 111980*x1^2*x2^3*x4^3 + 40412*x1^2*x2^2*x3^4
> + 154704*x1^2*x2^2*x3^2*x4^2 + 40412*x1^2*x2^2*x4^4
> + 2240*x1^2*x2*x3^5 - 23440*x1^2*x2*x3^4*x4
> - 23440*x1^2*x2*x3*x4^4 + 2240*x1^2*x2*x4^5
> - 6*x1^2*x3^6 - 2240*x1^2*x3^5*x4
> + 40412*x1^2*x3^4*x4^2 - 111980*x1^2*x3^3*x4^3
> + 40412*x1^2*x3^2*x4^4 - 2240*x1^2*x3*x4^5
> - 6*x1^2*x4^6 - 60*x1*x2^6*x3 + 60*x1*x2^6*x4
> + 2240*x1*x2^5*x3^2 - 2240*x1*x2^5*x4^2
> + 10180*x1*x2^4*x3^3 - 23440*x1*x2^4*x3^2*x4
> + 23440*x1*x2^4*x3*x4^2 - 10180*x1*x2^4*x4^3
> + 10180*x1*x2^3*x3^4 - 10180*x1*x2^3*x4^4
> + 2240*x1*x2^2*x3^5 + 23440*x1*x2^2*x3^4*x4
> - 23440*x1*x2^2*x3*x4^4 - 2240*x1*x2^2*x4^5
> - 60*x1*x2*x3^6 + 23440*x1*x2*x3^4*x4^2
> - 23440*x1*x2*x3^2*x4^4 + 60*x1*x2*x4^6 - 60*x1*x3^6*x4
> - 2240*x1*x3^5*x4^2 + 10180*x1*x3^4*x4^3
> - 10180*x1*x3^3*x4^4 + 2240*x1*x3^2*x4^5
> + 60*x1*x3*x4^6 + x2^8 - 6*x2^6*x3^2 - 60*x2^6*x3*x4
> - 6*x2^6*x4^2 - 2240*x2^5*x3^2*x4 - 2240*x2^5*x3*x4^2
> - 14*x2^4*x3^4 + 10180*x2^4*x3^3*x4
> + 40412*x2^4*x3^2*x4^2 + 10180*x2^4*x3*x4^3
> - 14*x2^4*x4^4 - 10180*x2^3*x3^4*x4
> - 111980*x2^3*x3^3*x4^2 - 111980*x2^3*x3^2*x4^3
> - 10180*x2^3*x3*x4^4 - 6*x2^2*x3^6 + 2240*x2^2*x3^5*x4
> + 40412*x2^2*x3^4*x4^2 + 111980*x2^2*x3^3*x4^3
> + 40412*x2^2*x3^2*x4^4 + 2240*x2^2*x3*x4^5 - 6*x2^2*x4^6
> + 60*x2*x3^6*x4 - 2240*x2*x3^5*x4^2 - 10180*x2*x3^4*x4^3
> - 10180*x2*x3^3*x4^4 - 2240*x2*x3^2*x4^5 + 60*x2*x3*x4^6
> + x3^8 - 6*x3^6*x4^2 - 14*x3^4*x4^4 - 6*x3^2*x4^6 + x4^8;

```

1. Degree 8

As explained above, degree 8 is sufficiently small so that all computations can be performed directly over the field K .

```
> K<z>:=CyclotomicField(12);
> P3K<x>:=ProjectiveSpace(K,3);
> time Zg:=Surface(P3K,Evaluate(g,[x[1],x[2],x[3],x[4]]));
Time: 11.040
> Zgsing:=SingularSubscheme(Zg);
> Dimension(Zgsing);
0
> unionhyp:=Scheme(P3K,x[1]*x[2]*x[3]*x[4]);
> Dimension(Intersection(Zgsing,unionhyp));
-1
```


Note that the fact that $\dim \mathcal{Z}(g)_{\text{sing}} = 0$ implies that $\mathcal{Z}(g)$ is geometrically irreducible. Note also that the last command implies that all the singular points are contained (for instance) in the affine chart $\mathcal{Z}(g)^{\text{aff}}$ defined by “ $x_4 \neq 0$ ”. We will make all the remaining computations in this affine chart:

```
> Zgaff:=AffinePatch(Zg,4);
> Zgaffsing:=SingularSubscheme(Zgaff);
> irr:=IrreducibleComponents(Zgaffsing);
> irr:=[ReducedSubscheme(i) : i in irr];
> Set([Degree(i) : i in irr]);
{ 1 }
> # irr;
44
```

The last two commands shows that all singular points have coordinates in K and that there are 44 singular points. We now check that they are quotient singularities of type D_4 :

```
> sings:=[RationalPoints(i) : i in irr];
> sings:=&cat [[Coordinates(i) : i in j] : j in sings];
> sings:=[Zgaff ! i : i in sings];
> test:=[ ];
> for i in sings
for>   do a,b,c:=IsSimpleSurfaceSingularity(i);
for>   test:=test cat [[* a,b,c *]];
for> end for;
> test[1];
[* true, D, 4 *]
> Set([i eq test[1] : i in test]);
{ true }
```

This concludes the proof of statement (a) of the Theorem. Note that the surface $\mathcal{Z}(g)$ has only 8 real points: Figure 1 shows part of the real locus of $\mathcal{Z}(g)$.

FIGURE 1. Part of the real locus of $\mathcal{Z}(g)$

2. Degree $8k$

Let U denote the open subset of $\mathbf{P}^3(\mathbb{C})$ defined by $x_1 x_2 x_3 x_4 \neq 0$ and let $\sigma_k : \mathbf{P}^3(\mathbb{C}) \rightarrow \mathbf{P}^3(\mathbb{C})$, $[x_1; x_2; x_3; x_4] \mapsto [x_1^k; x_2^k; x_3^k; x_4^k]$. The restriction of σ_k to a morphism $U \rightarrow U$ is an étale Galois covering, with group $(\mu_k)^4 / \Delta \mu_k$ (here, μ_k denotes the group of k -th roots of unity in $\mathbb{C}^\times$ and $\Delta : \mu_k \hookrightarrow (\mu_k)^4$ is the diagonal embedding). We have $\mathcal{Z}(g[k]) = \sigma_k^*(\mathcal{Z}(g))$.

Let us first prove that $\mathcal{Z}(g[k])$ is irreducible. We may assume that $k \geq 2$, as the result has been proved for $k = 1$ in the previous section. Note the following fact: if $i \neq j$, then the scheme $\mathcal{Z}_{i,j}$ defined by the ideal $\langle g, \frac{\partial g}{\partial x_i}, \frac{\partial g}{\partial x_j} \rangle$ has dimension 0:

```
> dg := [Derivative(g, i) : i in [1..4]];
> Dimension(Scheme(P3, [g, dg[1], dg[2]]));
0
> Dimension(Scheme(P3, [g, dg[1], dg[3]]));
0
> Dimension(Scheme(P3, [g, dg[1], dg[4]]));
0
> Dimension(Scheme(P3, [g, dg[2], dg[3]]));
0
> Dimension(Scheme(P3, [g, dg[2], dg[4]]));
0
> Dimension(Scheme(P3, [g, dg[3], dg[4]]));
0
```

Recall that

$$\frac{\partial g[k]}{\partial x_i} = k x_i^{k-1} \left(\frac{\partial g}{\partial x_i} \circ \sigma_k \right),$$

so the singular locus of $\mathcal{Z}(g[k])$ is contained in

$$\{p_1, p_2, p_3, p_4\} \cup \left(\bigcup_{i \neq j} \sigma_k^*(\mathcal{Z}_{i,j}) \right),$$

where $p_i = [\delta_{i1}; \delta_{i2}; \delta_{i3}; \delta_{i4}]$ (and δ_{ij} is the Kronecker symbol). Since σ_k is finite, this implies that $\mathcal{Z}(g[k])_{\text{sing}}$ has dimension 0, so $\mathcal{Z}(g[k])$ is irreducible.

Now, as $\sigma_k : U \rightarrow U$ is étale and the singular locus of $\mathcal{Z}(g)$ is contained in U (see the command `Dimension(Intersection(Zgsing, unionhyp))` of the previous section), the 44 singularities of $\mathcal{Z}(g)$ give $44k^3$ singularities in $\mathcal{Z}(g[k])$ of the same type, i.e. quotient singularities of type D_4 . This proves the statement (b) of the Theorem.

Note that, for $k = 2, 3$ and 4 (and maybe for bigger k) we will prove in the next sections that $\mathcal{Z}(g[k])$ contains singular points outside of U .

3. Degree 16

Using the morphism σ_2 defined in the previous section and the fact that it is étale over U , we get that $\mathcal{Z}(g[2]) \cap U$ has exactly 352 singular points, which are all quotient singularities of type D_4 . We now need to determine the singularities which are not contained in U . So let H be the complement of U in $\mathbb{P}^3(\mathbb{C})$.

```
> g2:=Evaluate(g, [x1^2, x2^2, x3^2, x4^2]);
> Z2:=Surface(P3, g2);
> Z2sing:=SingularSubscheme(Z2);
> time Dimension(Z2sing);
0
Time: 20.920
> unionhyp:=Scheme(P3, x1*x2*x3*x4);
> Z2singhyp:=Intersection(Z2sing, unionhyp);
> Z2singhyp:=ReducedSubscheme(Z2singhyp);
> irrsinghyp:=IrreducibleComponents(Z2singhyp);
> # irrsinghyp;
18
> Set([Degree(i) : i in irrsinghyp]);
{ 4, 8 }
> &+ [Degree(i) : i in irrsinghyp];
120
```

This shows that $\mathcal{Z}(g[2])_{\text{sing}} \cap H$ contains 120 points, none of which being defined over $\mathbb{Q}$. We denote by a an 8-th root of unity and by K_4 the group the cyclotomic field $\mathbb{Q}(a)$. We denote by K_8 the splitting field of the polynomial

$$T^{16} + 3248T^{12} + 23100000T^8 + 20300000000T^4 + 3906250000000,$$

and by u a root of this polynomial (one can check that K_8 is a Galois extension of $\mathbb{Q}$).

It turns out that every singular point which is defined over a field extension of degree 4 (resp. 8) has coordinates in K_4 (resp. K_8). We separate the treatment of these two families.

Singular points defined over K_4 . We will check that all of them are quotient singularities of type A_1 :

```
> singhyp4:=[i : i in irrsinghyp | Degree(i) eq 4];
> K4<a>:=CyclotomicField(8);
> Z2K4:=ChangeRing(Z2, K4);
> P3K4<x>:=AmbientSpace(Z2K4);
> singhyp4:=[ChangeRing(i, K4) : i in singhyp4];
> singhyp4:=[* [j : j in RationalPoints(i)]
> : i in singhyp4 *];
> singhyp4:=[Coordinates(i[1]) : i in singhyp4];
> singhyp4:=[Z2K4 ! i : i in singhyp4];
```

```

> types:=[];
> for i in singhyp4 do
for>   te,ty,ra:=IsSimpleSurfaceSingularity(i);
for>   types:=types cat [[* te,ty,ra *]];
for> end for;
> types;
[ [* true, A, 1 *], [* true, A, 1 *],
  [* true, A, 1 *], [* true, A, 1 *],
  [* true, A, 1 *], [* true, A, 1 *] ]

```

We have chosen one point in each irreducible components over $\mathbb{Q}$, so this proves that $\mathcal{Z}(g)$ contains at least 24 quotient singularities of tpe A_1 .

Singular points defined over K_8 . We will check that all of them are quotient singularities of type A_2 . So let us first pick one point in each irreducible component over $\mathbb{Q}$:

```

> singhyp8:=[i : i in irrsinghyp | Degree(i) eq 8];
> POL<T>:=PolynomialRing(Q);
> K8<u>:=NumberField(T^16 + 3248*T^12
> + 23100000*T^8 + 20300000000*T^4
> + 39062500000000);
> Z2K8:=ChangeRing(Z2,K8);
> P3K8<[x]>:=AmbientSpace(Z2K8);
> singhyp8:=[ChangeRing(i,K8) : i in singhyp8];
> singhyp8:=[* [j : j in RationalPoints(i)]
> : i in singhyp8 *];
> singhyp8:=[Coordinates(i[1]) : i in singhyp8];
> singhyp8:=[Z2K8 ! i : i in singhyp8];


```

The Magma command `IsSimpleSurfaceSingularity` does not answer in a reasonable amount of time. Let us explain how we proceed for one of the points in the list `singhyp8`. So let p be the first point in the list. It lives in the affine chart defined by “ $x_3 \neq 0$ ”, and we denote by (x, y, z) the variables $(x_1/x_3, x_2/x_3, x_4/x_3)$. Let $f(x, y, z)$ the defining polynomial of $\mathcal{Z}(g[2])$ in this affine chart, after translating p at the origin $(0, 0, 0)$. We denote by f_2 its homogeneous component of degree 2:

```

> A3K8<x,y,z>:=AffineSpace(K8,3);
> p:=singhyp8[1];
> cop:=Coordinates(p);
> f:=Evaluate(g2,[x+cop[1],y+cop[2],1,z+cop[4]]);
> cof:=Coefficients(f);
> mof:=Monomials(f);
> l:=# mof;
> f2:=&+ [cof[i]*mof[i] : i in [1..l] |
> Degree(mof[i]) eq 2];
> f2:=f2/Reverse(Coefficients(f2))[1];
> Factorization(f2-z^2);
[
  <x + 1/264000000000000*(-57*u^14 - 5500*u^12
  - 372636*u^10 - 17864000*u^8
  - 1456950000*u^6 - 92675000000*u^4
  - 3075225000000*u^2 -
  55825000000000)*y, 2>
]

```


FIGURE 2. Part of the real locus of $\mathcal{Z}(g[2])$

The last command shows that, up to a change of coordinates, f_2 might be transformed to $y^2 + z^2$. By standard arguments, this proves that p is a quotient singularity of type A_k , for some $k \geq 2$, which can be obtained as the Milnor number of $\mathcal{Z}(g[2])$ at p : however, Magma cannot compute this Milnor number in a reasonable amount of time and we need to copy the polynomial f in the software Singular [DGPS] to compute this Milnor number (!): we obtain 2. So p is a quotient singularity of type A_2 .

Repeating 12 times the same method (we omit details), we get that $\mathcal{Z}(g[2])$ has 96 quotient singularities of type A_2 . This proves statement (c) of the Theorem.

Figure 2 shows part of the real locus of $\mathcal{Z}(g)$.

4. Degree 24

We now come to the surface $\mathcal{Z}(g[3])$. For shortening the first computations, we will define it over $\mathbb{Q}$.

```
> g3:=Evaluate(g, [x1^3, x2^3, x3^3, x4^3]);
> Zg3:=Surface(P3, g3);
> Zg3sing:=SingularSubscheme(Zg3);
> time Dimension(Zg3sing);
0
Time: 13.210
```

Again, the fact that the $\dim \mathcal{Z}(g[3])_{\text{sing}} = 0$ implies that $\mathcal{Z}(g[3])$ is geometrically irreducible. We will compute the singular locus of $\mathcal{Z}(g[3])$ over $\mathbb{Q}$, by separating its intersection with the affine chart “ $x_1 \neq 0$ ” and its intersection with the hyperplane “ $x_1 = 0$ ”.

```
> Zg3aff:=AffinePatch(Zg3, 1);
> Zg3affsing:=SingularSubscheme(Zg3aff);
> time irr3aff:=IrreducibleComponents(Zg3affsing);
Time: 48.370
> irr3aff:=[ReducedSubscheme(i) : i in irr3aff];
> hyp:=Intersection(Zg3sing, Scheme(P3, x1));
> irrhyp:=IrreducibleComponents(hyp);
> irrhyp:=[ReducedSubscheme(i) : i in irrhyp];
```

We now check that all the singular points are rational over the cyclotomic field $K = \mathbb{Q}(z)$:

```
> K<z>:=CyclotomicField(12);
> irr3aff:=[ChangeRing(i,K) : i in irr3aff];
> irr3aff:=&cat [IrreducibleComponents(i) : i in irr3aff];
> irrhyp:=[ChangeRing(i,K) : i in irrhyp];
> irrhyp:=&cat [IrreducibleComponents(i) : i in irrhyp];
> Set([Degree(i) : i in irr3aff]);
{ 1 }
> Set([Degree(i) : i in irrhyp]);
{ 1 }
> (# irr3aff) + (# irrhyp);
1440
```

The above last command shows that $\mathcal{Z}(g[3])$ has exactly 1440 singular points. We now pick a singular point p in $\mathcal{Z}(g[3])$, and show that it is a quotient singularity of type D_4 :

```
> Zg3K:=ChangeRing(Zg3,K);
> p:= Zg3K ! [0,-(2*z^3-z^2-z+1), (z^3+z^2+z), 2];
> IsSingular(Zg3K,p);
true
> IsSimpleSurfaceSingularity(p);
true D 4
```

We could prove that all other singular points are quotient singularities of type D_4 (by making 1440 computations!), but we prefer to do it using group theory. So let W be the complex reflection group denoted by G_{32} in Shephard-Todd classification (see for instance [LeTa]). We could obtain it by the MAGMA command $W := \text{ShephardTodd}(32)$, but it turns out that the group implemented in this way in MAGMA does not stabilize $g[3]$: with the version of G_{32} implemented in MAGMA, we would need to replace $g[3]$ by a much more complicated polynomial with huge coefficients lying in the cyclotomic field $\mathbb{Q}(z^4)$. This would have lead to very long computations (we could even not check that the 1440 singular points are the only ones: MAGMA computations do not conclude, at least after few hours). We prefer to use the model implemented by Jean Michel [Mic] in the CHEVIE package of GAP3.

```
> a:=z^4;
> s1:=Matrix(K,4,4,
> [ [ 1, 0, 0, 0 ],
> [ 0, 1, 0, 0 ],
> [ 0, 0, a, 0 ],
> [ 0, 0, 0, 1 ] ]);
>
> s2:=Matrix(K,4,4,
> [ [-1/3*a-2/3*a^2, 2/3*a+1/3*a^2, 2/3*a+1/3*a^2, 0 ],
> [ 2/3*a+1/3*a^2, -1/3*a-2/3*a^2, 2/3*a+1/3*a^2, 0 ],
> [ 2/3*a+1/3*a^2, 2/3*a+1/3*a^2, -1/3*a-2/3*a^2, 0 ],
> [ 0, 0, 0, 1 ] ]);
>
> s3:=Matrix(K,4,4,
> [ [ 1, 0, 0, 0 ],
> [ 0, a, 0, 0 ],
> [ 0, 0, 1, 0 ],
> [ 0, 0, 0, 1 ] ]);
>
> s4:=Matrix(K,4,4,
```

```

> [ [-1/3*a-2/3*a^2, -2/3*a-1/3*a^2, 0, -2/3*a-1/3*a^2 ],
>   [-2/3*a-1/3*a^2, -1/3*a-2/3*a^2, 0, 2/3*a+1/3*a^2 ],
>   [ 0, 0, 1, 0 ],
>   [-2/3*a-1/3*a^2, 2/3*a+1/3*a^2, 0, -1/3*a-2/3*a^2 ] ] );
>
> W:=MatrixGroup<4,K | [s1,s2,s3,s4]>;
> Order(W);
155520
> Order(W/Centre(W));
25920
> test:=IsIsomorphic(W, ShephardTodd(32));
> test;
true

```

We then check that $g[3]$ is invariant under the action of W :

```

> R:=InvariantRing(W);
> P<X1,X2,X3,X4>:=PolynomialRing(R);
> g3copy:=Evaluate(g3, [X1,X2,X3,X4]);
> [g3copy^w eq g3copy : w in Generators(W)];
[ true, true, true, true ]

```

This shows that the group W defined as above, which is a complex reflection group of type G_{32} in Shephard-Todd classification, acts on $\mathcal{Z}$. Its center $Z(W)$ has order 6, so $W/Z(W)$ has order 25920 and acts faithfully on $\mathcal{Z}$. We conclude by computing the W -orbit of the point p . We do this by first computing in the vector space $V \simeq \mathbb{C}^4$ and then projecting on $\mathbf{P}(V) \simeq \mathbf{P}^3(\mathbb{C})$:

```

> P3K:=ProjectiveSpace(K, 3);
> ZW:=Centre(W);
> rep:=Transversal(W, ZW);
> V:=VectorSpace(W);
> v:= V ! Coordinates(p);
> orbv:=[v^(Transpose(w)) : w in rep];
> orbp:=[P3K ! Coordinates(V,i) : i in orbv];
> orbp:=Set(orbp);
> # orbp;
1440


```

As the W -orbit of p has 1440 elements and as $\mathcal{Z}(g[3])$ has 1440 singular points, the proof of statement (b) of the Theorem is complete. Note also that, in the given model, the surface $\mathcal{Z}(g[3])$ has only 32 real points: Figure 3 gives partial views of its real locus.

Remark 3. From Section 2, we deduce that $\mathcal{Z}(g[4])_{\text{sing}}$ has 2816 quotient singularities of type D_4 in the open subset U and it can be checked that it has 480 other singular points outside of U , for which we did not determine the type. ■

Remark 4. After investigations in the invariant rings of several complex reflection groups, we have also been able to construct curves with many singularities. For example:

- Using the reflection group $W = G_{24}$, we have obtained a cuspidal curve of degree 14 in $\mathbf{P}^2(\mathbb{C})$ with exactly 42 cusps (all lying in a single W -orbit). Note that $W/Z(W)$ has order 168 and is isomorphic to $\mathbf{GL}_3(\mathbb{F}_2)$.
- Using the reflection group $W = G_{26}$, we have obtained a curve of degree 18 in $\mathbf{P}^2(\mathbb{C})$ with 72 cusps and 12 nodes (these are the two W -orbits of singular points). Note that $W/Z(W)$ has order 216.

FIGURE 3. Part of the real locus of $\mathcal{Z}(g[3])$

Also, other singular surfaces have been obtained. For example:

- Using the reflection group $W = G_{29}$ (note that $W/Z(W)$ has order 1920), we have obtained:
 - a surface of degree 8 in $\mathbf{P}^3(\mathbb{C})$ with 160 nodes, all belonging to the same W -orbit. Recall that *Endraß octic surface* [End] has degree 8 and 168 nodes and its automorphism group has order 16.
 - a surface of degree 8 in $\mathbf{P}^3(\mathbb{C})$ with 20 singular points of multiplicity 3 and Milnor number 11, all belonging to the same W -orbit.
- Using the reflection group $W = G_{31}$ (note that $W/Z(W)$ has order 11520), we have obtained a surface of degree 20 in $\mathbf{P}^3(\mathbb{C})$ with 1920 nodes, all lying in the same W -orbit. Recall that *Chmutov surface* [Chm] of degree 20 has 2926 nodes.

Details will appear in a forthcoming paper. ■

Acknowledgements. The hidden computations which lead to the discovery of the polynomial g were done using the High Performance Computing facilities of the MSRI, during the visit of the author in Spring 2018.

I wish to thank warmly Alessandra Sarti, Oliver Labs and Duco van Straten for useful comments and references.

Figures were realized using the software SURFER [Sur].

References

- [Bar] W. BARTH, *Two projective surfaces with many nodes, admitting the symmetries of the icosahedron*, J. Algebraic Geom. **5** (1996), 173-186.
- [Chm] S. V. CHMUTOV, *Examples of projective surfaces with many singularities*, J. Algebraic Geom. **1** (1992), 191-196.
- [DGPS] W. DECKER, G.-M. GREUEL, G. PFISTER & H. SCHÖNEMANN, *Singular 4-1-1 — A computer algebra system for polynomial computations*, <http://www.singular.uni-kl.de> (2018).

- [End] S. ENDRAß, *A projective surface of degree eight with 168 nodes*, J. Algebraic Geom. **6** (1997), 325-334.
- [LeTa] G. I. LEHRER & D. E. TAYLOR, *Unitary reflection groups*, Australian Mathematical Society Lecture Series **20**. Cambridge University Press, Cambridge, 2009. viii+294 pp.
- [Magma] W. BOSMA, J. CANNON & C. PLAYOUST, *The Magma algebra system. I. The user language*, J. Symbolic Comput. **24** (1997), 235-265.
- [Mic] J. MICHEL, *The development version of the CHEVIE package of GAP3*, J. of Algebra **435** (2015), 308-336.
- [Miy] Y. MIYAOKA, *The maximal number of quotient singularities on surfaces with given numerical invariants*, Math. Ann. **268** (1984), 159-171.
- [Sar1] A. SARTI, *Pencils of symmetric surfaces in $\mathbf{P}^3$* , J. of Algebra **246** (2001), 429-452.
- [Sar2] A. SARTI, *Symmetric surfaces with many singularities*, Comm. in Alg. **32** (2004), 3745-3770.
- [Sar3] A. SARTI, *Symmetrische Flächen mit gewöhnlichen Doppelpunkten*, Math. Semesterber. **55** (2008), 1-5.
- [Sur] www.imaginary.org/program/surfer.

April 23, 2018

CÉDRIC BONNAFÉ, Institut Montpellierain Alexander Grothendieck (CNRS: UMR 5149), Université Montpellier 2, Case Courrier 051, Place Eugène Bataillon, 34095 MONTPELLIER Cedex, FRANCE
E-mail: cedric.bonnafe@umontpellier.fr